- **By Darboux lemma we may assume that the symplectic form has the form**

  ω = dx^1 ∧ dx^3 + dx^2 ∧ dx^4

  in local coordinates (x^1, x^2, x^3, x^4). Then

  ω = 1/2 OmegaForm(i,j) dx^i ∧ dx^j.

```
OmegaForm = {
   {0, 0, 1, 0},
   {0, 0, 0, 1},
   {-1, 0, 0, 0},
   {0, -1, 0, 0}
  };

OmInv = Inverse[OmegaForm];
OmInv // MatrixForm
```

$$\begin{pmatrix} 0 & 0 & -1 & 0 \\ 0 & 0 & 0 & -1 \\ 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \end{pmatrix}$$

```
(* Xi is an arbitrary 2-form. *)
XiNorm[a_, b_] := Signature[{a, b}] ξ[Min[{a, b}], Max[{a, b}]]

(* evaluate
        dx^i1 ∧ ... ∧ dx^i4( d/dx^k1 ∧ ... ∧ d/dx^k4)
*)
Perm[i1_, i2_, i3_, i4_, k1_, k2_, k3_, k4_] := Module[
  {Ivec, Kvec, a, b},
  Ivec = {i1, i2, i3, i4};
  Kvec = {k1, k2, k3, k4};
  Det[Table[KroneckerDelta[Ivec[[a]], Kvec[[b]]], {a, 1, 4}, {b, 1, 4}]]
 ]
```

- **Evaulate traditional definition in coordinates for a 2-form on a 4-manifold.**
  See Libermann and Marle: Symplectic geometry and analytic mechanics, p. 43

- **Evaluate 1/2 ω vedge ω(ξ^sharp wedge d/dx^k3 ∧ d/dx^k4)**

```
defTrad = Table[
   Sum[1 / 16 OmegaForm[[i1]][[i2]]
     OmegaForm[[i3]][[i4]] XiNorm[l, m] OmInv[[l]][[k1]] OmInv[[m]][[k2]]
     Perm[i1, i2, i3, i4, k1, k2, k3, k4],
    {i1, 1, 4}, {i2, 1, 4}, {i3, 1, 4}, {i4, 1, 4},
    {l, 1, 4}, {m, 1, 4}, {k1, 1, 4}, {k2, 1, 4}
   ],
   {k3, 1, 4}, {k4, 1, 4}
  ];
defTrad // MatrixForm
```

$$\begin{pmatrix} 0 & -\xi[1,2] & \xi[2,4] & -\xi[1,4] \\ \xi[1,2] & 0 & -\xi[2,3] & \xi[1,3] \\ -\xi[2,4] & \xi[2,3] & 0 & -\xi[3,4] \\ \xi[1,4] & -\xi[1,3] & \xi[3,4] & 0 \end{pmatrix}$$

- **Definition in paper:**

ast(xi)_ij = -1/2 ω^ab xi_ab ω_ij - xi_ij

```
defPaper = Table[
   -1 / 2 Sum[OmInv[[a]][[b]] XiNorm[a, b], {a, 1, 4}, {b, 1, 4}]
     OmegaForm[[i]][[j]] - XiNorm[i, j],
   {i, 1, 4}, {j, 1, 4}
  ];
defPaper = Simplify[defPaper];
defPaper // MatrixForm
```

$$\begin{pmatrix} 0 & -\xi[1,2] & \xi[2,4] & -\xi[1,4] \\ \xi[1,2] & 0 & -\xi[2,3] & \xi[1,3] \\ -\xi[2,4] & \xi[2,3] & 0 & -\xi[3,4] \\ \xi[1,4] & -\xi[1,3] & \xi[3,4] & 0 \end{pmatrix}$$

- **Check that the two definitions coincide**

```
Simplify[defPaper == defTrad]
```

```
True
```

- **Note: The above computation also gives an explicit expression for the star operator in Darboux coordinates.**